Derivata

```
(* Derivate delle funzioni elementari *)
```

In[1]:=

```
RapportoIncrementale[esp_,x_,h_] := ((esp /. x -> x + h) - esp)/h

  (* in questa definizione è necessario utilizzare l'operatore
     := di assegnazione con valutazione differita ... perché? *)
```

In[4]:=

```
RapportoIncrementale[c,x,h]
```

Out[4]=

```
0
```

In[5]:=

```
Derivata[c_,x_] /; FreeQ[c,x] = 0
```

Out[5]=

```
0
```

In[6]:=

```
RapportoIncrementale[x,x,h]
```

Out[6]=

```
1
```

In[7]:=

```
Derivata[x_,x_] = 1
```

Out[7]=

```
1
```

In[8]:=

```
RapportoIncrementale[Exp[x],x,h]
```

Out[8]=

$$\frac{-E^{x} + E^{h + x}}{h}$$

In[9]:=

```
Simplify[%]
```

Out[9]=

$$\frac{E^{x} (-1 + E^{h})}{h}$$

In[10]:=

```
Derivata[Exp[x_],x_] = % /. (E^h - 1)/h -> 1 (* limite notevole *)
```

Out[10]=

$$E^{x}$$

In[11]:=
```
RapportoIncrementale[Sin[x],x,h]
```

Out[11]=
$$\frac{-\text{Sin}[x] + \text{Sin}[h + x]}{h}$$

In[12]:=
```
Simplify[%]
```

Out[12]=
$$\frac{2\ \text{Cos}[\frac{h}{2} + x]\ \text{Sin}[\frac{h}{2}]}{h}$$

In[13]:=
```
% /. 2/h Sin[h/2] -> 1 (* limite notevole *)
```

Out[13]=
$$\text{Cos}[\frac{h}{2} + x]$$

In[14]:=
```
Derivata[Sin[x_],x_] = % /. h -> 0 (* ora si può sostituire *)
```

Out[14]=
$$\text{Cos}[x]$$

In[15]:=
```
RapportoIncrementale[Cos[x],x,h]
```

Out[15]=
$$\frac{-\text{Cos}[x] + \text{Cos}[h + x]}{h}$$

In[16]:=
```
Simplify[%]
```

Out[16]=
$$\frac{-2\ \text{Sin}[\frac{h}{2}]\ \text{Sin}[\frac{h}{2} + x]}{h}$$

In[17]:=
```
% /. -2/h Sin[h/2] -> -1 (* limite notevole *)
```

Out[17]=
$$-\text{Sin}[\frac{h}{2} + x]$$

In[18]:=
```
Derivata[Cos[x_],x_] = % /. h -> 0 (* ora si può sostituire *)
```

Out[18]=
$$-\text{Sin}[x]$$

```
(* Regole di derivazione *)
```

In[19]:=

```
Derivata[e1_ + e2_,x_] := Derivata[e1,x] + Derivata[e2,x]

Derivata[e1_ e2_,x_] := Derivata[e1,x] e2 + e1 Derivata[e2,x]

Derivata[e1_^e2_,x_] := e1^(e2 - 1) e2 Derivata[e1,x] +
                        e1^e2 Log[e1] Derivata[e2,x]

Derivata[f_[e_],x_] /; e =!= x := Derivata[f[e],e] Derivata[e,x]

(* verifichiamo ... *)
```

In[24]:=

```
Derivata[c f[x],x]
```

Out[24]=

```
c Derivata[f[x], x]
```

In[25]:=

```
Derivata[f[x] + g[x],x]
```

Out[25]=

```
Derivata[f[x], x] + Derivata[g[x], x]
```

In[26]:=

```
Derivata[f[x] - g[x],x]
```

Out[26]=

```
Derivata[f[x], x] - Derivata[g[x], x]
```

In[27]:=

```
Derivata[f[x] g[x],x]
```

Out[27]=

```
Derivata[g[x], x] f[x] + Derivata[f[x], x] g[x]
```

In[28]:=

```
Derivata[f[x]/g[x],x]
```

Out[28]=

$$-\left(\frac{\text{Derivata}[g[x],\ x]\ f[x]}{g[x]^2}\right) + \frac{\text{Derivata}[f[x],\ x]}{g[x]}$$

In[29]:=

```
Together[%]
```

Out[29]=

$$\frac{-(\text{Derivata}[g[x],\ x]\ f[x]) + \text{Derivata}[f[x],\ x]\ g[x]}{g[x]^2}$$

In[30]:=

```
Derivata[f[x]^g[x],x]
```

Out[30]=

$$\text{Derivata}[f[x], x]\ f[x]^{-1+g[x]}\ g[x] + \text{Derivata}[g[x], x]\ f[x]^{g[x]}\ \text{Log}[f[x]]$$

In[31]:=

```
Derivata[f[g[x]],x]
```

Out[31]=

$$\text{Derivata}[f[g[x]], g[x]]\ \text{Derivata}[g[x], x]$$

```
(* altre derivate di funzioni elementari *)
```

In[32]:=

```
Derivata[f[x]/g[x],x] /. {f -> Sin,g -> Cos}
```

Out[32]=

$$1 + \text{Tan}[x]^2$$

In[33]:=

```
Derivata[Tan[x_],x_] = %
```

Out[33]=

$$1 + \text{Tan}[x]^2$$

In[34]:=

```
Derivata[f[g[x]],x] /. {f -> Log,g -> Exp}
```

Out[34]=

$$E^x\ \text{Derivata}[\text{Log}[E^x], E^x]$$

In[35]:=

```
% /. E^x -> y
```

Out[35]=

$$y\ \text{Derivata}[\text{Log}[y], y]$$

In[36]:=

```
Solve[% == 1,Derivata[Log[y],y]]
```

Out[36]=

$$\left\{\left\{\text{Derivata}[\text{Log}[y], y] \to \frac{1}{y}\right\}\right\}$$

In[37]:=

```
Derivata[Log[y_],y_] = Derivata[Log[y],y] /. %[[1]]
```

Out[37]=

$$\frac{1}{y}$$

In[38]:=

```
Derivata[f[g[x]],x] /. {f -> ArcSin,g -> Sin}
```

Out[38]=

```
Cos[x] Derivata[ArcSin[Sin[x]], Sin[x]]
```

In[39]:=

```
% /. {Sin[x] -> y,Cos[x] -> Sqrt[1 - y^2]}
```

Out[39]=

```
                    2
Sqrt[1 - y ] Derivata[ArcSin[y], y]
```

In[40]:=

```
Solve[% == 1,Derivata[ArcSin[y],y]]
```

Out[40]=

```
                                  1
{{Derivata[ArcSin[y], y] -> ------------}}
                                   2
                            Sqrt[1 - y ]
```

In[41]:=

```
Derivata[ArcSin[y_],y_] = Derivata[ArcSin[y],y] /. %[[1]]
```

Out[41]=

```
      1
------------
           2
Sqrt[1 - y ]
```

In[42]:=

```
Derivata[f[g[x]],x] /. {f -> ArcCos,g -> Cos}
```

Out[42]=

```
-(Derivata[ArcCos[Cos[x]], Cos[x]] Sin[x])
```

In[43]:=

```
% /. {Cos[x] -> y,Sin[x] -> Sqrt[1 - y^2]}
```

Out[43]=

```
                    2
-(Sqrt[1 - y ] Derivata[ArcCos[y], y])
```

In[44]:=

```
Solve[% == 1,Derivata[ArcCos[y],y]]
```

Out[44]=

```
                                   1
{{Derivata[ArcCos[y], y] -> -(------------)}}
                                    2
                             Sqrt[1 - y ]
```

In[45]:=

```
Derivata[ArcCos[y_],y_] = Derivata[ArcCos[y],y] /. %[[1]]
```

Out[45]=

```
       1
-(------------)
            2
  Sqrt[1 - y ]
```

In[46]:=

```
Derivata[f[g[x]],x] /. {f -> ArcTan,g -> Tan}
```

Out[46]=

$$\text{Derivata}[\text{ArcTan}[\text{Tan}[x]],\ \text{Tan}[x]]\ (1 + \text{Tan}[x]^2)$$

In[47]:=

```
% /. Tan[x] -> y
```

Out[47]=

$$(1 + y^2)\ \text{Derivata}[\text{ArcTan}[y],\ y]$$

In[48]:=

```
Solve[% == 1,Derivata[ArcTan[y],y]]
```

Out[48]=

$$\{\{\text{Derivata}[\text{ArcTan}[y],\ y] \to \frac{1}{1 + y^2}\}\}$$

In[49]:=

```
Derivata[ArcTan[y_],y_] = Derivata[ArcTan[y],y] /. %[[1]]
```

Out[49]=

$$\frac{1}{1 + y^2}$$

```
(* un esempio di calcolo ... *)
```

In[50]:=

```
Sqrt[Log[x]/(x^2 + 1) Sin[x]^(Cos[x] + E^x)]
```

Out[50]=

$$\text{Sqrt}\left[\frac{\text{Log}[x]\ \text{Sin}[x]^{E^x + \text{Cos}[x]}}{1 + x^2}\right]$$

In[51]:=

```
Derivata[%,x]
```

Out[51]=

$$\left(\frac{-2\,x\,\text{Log}[x]\,\text{Sin}[x]^{E^x+\text{Cos}[x]}}{(1+x^2)^2}+\left(\frac{\text{Sin}[x]^{E^x+\text{Cos}[x]}}{x}+\text{Log}[x]\right)\left(\text{Cos}[x]\,(E^x+\text{Cos}[x])\,\text{Sin}[x]^{-1+E^x+\text{Cos}[x]}+\text{Log}[\text{Sin}[x]]\,(E^x-\text{Sin}[x])\,\text{Sin}[x]^{E^x+\text{Cos}[x]}\right)\right)\Big/(1+x^2)\Big/\left(2\,\text{Sqrt}\left[\frac{\text{Log}[x]\,\text{Sin}[x]^{E^x+\text{Cos}[x]}}{1+x^2}\right]\right)$$

In[52]:=

```
Integrate[%,x]
```

Out[52]=

$$\text{Sqrt}\left[\frac{\text{Log}[x]\,\text{Sin}[x]^{E^x+\text{Cos}[x]}}{1+x^2}\right]$$